Effective CMake

a random seletion of best practices

Daniel Pfeifer

May 19, 2017

daniel@pfeifer-mail.de

Opening

Why?

The way you use CMake affects your users!

CMake's similarities with C++

- · big userbase, industry dominance
- focus on backwards compatibility
- · complex, feature rich, "multi paradigm"
- · bad reputation, "bloated", "horrible syntax"
- · some not very well known features

Standards

HOW STANDARDS PROLIFERATE: (SEE: A/C CHARGERS, CHARACTER ENCODINGS, INSTANT MESSAGING, ETC.)

SITUATION: THERE ARE 14 COMPETING STANDARDS. IH?! RIDICULOUS!
WE NEED TO DEVELOP
ONE UNIVERSAL STANDARD
THAT COVERS EVERYONE'S
USE CASES. YEAH!

SXXIII
SITUATION:
THERE ARE
15 COMPETING
STANDARDS.

Use the same principles for CMakeLists.txt

and modules as for the rest of your codebase.

CMake is code.

Language

Organization

- **Directories** that contain a **CMakeLists.txt** are the entry point for the build system generator. Subdirectories may be added with **add_subdirectory()** and must contain a **CMakeLists.txt** too.

 - Modules are <script>.cmake files located in the CMAKE_MODULE_PATH.

 Modules can be loaded with the include() command.

Commands

- command name(space separated list of strings)
 - · Scripting commands change state of command processor
 - · set variables
 - · change behavior of other commands
 - Project commands
 - · create build targets
 - · modify build targets
 - · Command invocations are not expressions.

Variables

```
set(hello world)
message(STATUS "hello, ${hello}")
```

- · Set with the set() command.
- Expand with \${}.
- · Variables and values are strings.
- · Lists are ;-separated strings.
- CMake variables are not environment variables (unlike Makefile).
- · Unset variable expands to empty string.

Comments

```
# a single line comment

# [==[

multi line comments

# [=[

may be nested

# #]=]

# #]==]
```

Generator expressions

```
target_compile_definitions(foo PRIVATE
VERBOSITY=$<IF:$<CONFIG:Debug>,30,10>"
)
```

- Generator expressions use the \$<> syntax.
- Not evaluated by command interpreter.
 It is just a string with \$<>.
- Evaluated during build system generation.
- · Not supported in all commands (obviously).

Custom Commands

Two types of commands

- · Commands can be added with function() or macro().
- · Difference is like in C++.
- When a new command replaces an existing command, the old one can be accessed with a _ prefix.

Custom command: Function

```
function(my_command input output)

# ...

set(${output} ... PARENT_SCOPE)

endfunction()

my_command(foo bar)
```

- Variables are scoped to the function, unless set with PARENT_SCOPE.
- Available variables: input, output, ARGC, ARGV, ARGN, ARG0, ARG1, ARG2, ...
- Example: **\${output}** expands to **bar**.

Custom command: Macro

```
macro(my_command input output)
    # ...
endmacro()
my_command(foo bar)
```

- · No extra scope.
- Text replacements: \${input}, \${output}, \${ARGC}, \${ARGV}, \${ARGN}, \${ARG0}, \${ARG1}, \${ARG2}, ...
- Example: **\${output}** is replaced by bar.

Create macros to wrap commands that have output parametes.

Otherwise, create a function.

Evolving CMake code

Deprecate CMake commands

```
macro(my_command)
macro(my_command)
message(DEPRECATION
    "The my_command command is deprecated!")
my_command(${ARGV})
endmacro()
```

Deprecate CMake variables

```
set(hello "hello world!")
2
3 function( deprecated var var access)
   if(access STREQUAL "READ ACCESS")
     message(DEPRECATION
        "The variable '${var}' is deprecated!")
   endif()
s endfunction()
10 variable watch(hello deprecated var)
```

Variables are so CMake 2.8.12.

Modern CMake is about Targets and Properties!

Targets and Properties

Look Ma, no Variables!

```
add_library(Foo foo.cpp)
target_link_libraries(Foo PRIVATE Bar::Bar)

if(WIN32)
target_sources(Foo PRIVATE foo_win32.cpp)
target_link_libraries(Foo PRIVATE Bar::Win32Support)
endif()
```

Avoid custom variables

in the arguments of project commands.

Don't use file(GLOB) in projects.

Imagine Targets as Objects

- · Constructors:
 - add_executable()
 - · add_library()
- · Member variables:
 - Target properties (too many to list here).
- · Member functions:
 - get_target_property()
 - · set_target_properties()
 - get_property(TARGET)
 - set property(TARGET)
 - target_compile_definitions()
 - target_compile_features()
 - target_compile_options()
 - target_include_directories()
 - target_link_libraries()
 - target_sources()

Forget those commands:

```
add_compile_options()
include_directories()
  link_directories()
  link libraries()
```

```
target_compile_features(Foo
PUBLIC
cxx_strong_enums
PRIVATE
cxx_lambdas
cxx_range_for
)
```

- Adds cxx_strong_enums to the target properties
 COMPILE_FEATURES and INTERFACE_COMPILE_FEATURES.
- Adds cxx_lambdas;cxx_range_for to the target property COMPILE_FEATURES.



Build Specification and Usage Requirements

- Non-INTERFACE_ properties define the build specification of a target.
- INTERFACE_ properties define the usage requirements of a target.

Build Specification and Usage Requirements

- PRIVATE populates the non-INTERFACE_ property.
- \cdot INTERFACE populates the INTERFACE_ property.
- PUBLIC populates both.

Use target_link_libraries()

to express direct dependencies!

```
target_link_libraries(Foo
PUBLIC Bar::Bar
PRIVATE Cow::Cow

)
```

- Adds Bar::Bar to the target properties LINK_LIBRARIES and INTERFACE_LINK_LIBRARIES.
- · Adds Cow::Cow to the target property LINK_LIBRARIES.

```
target_link_libraries(Foo
PUBLIC Bar::Bar
PRIVATE Cow::Cow

)
```

- Adds Bar::Bar to the target properties LINK_LIBRARIES and INTERFACE_LINK_LIBRARIES.
- · Adds Cow::Cow to the target property LINK_LIBRARIES.
- Effectively adds all INTERFACE_<property> of Bar::Bar to <property> and INTERFACE_<property>.
- Effectively adds all INTERFACE_<property> of Cow::Cow to <property>.

```
target_link_libraries(Foo
PUBLIC Bar::Bar
PRIVATE Cow::Cow

)
```

- Adds Bar::Bar to the target properties LINK_LIBRARIES and INTERFACE_LINK_LIBRARIES.
- Adds Cow::Cow to the target property LINK_LIBRARIES.
- Effectively adds all INTERFACE_<property> of Bar::Bar to
 <property> and INTERFACE_<property>.
- Effectively adds all INTERFACE_<property> of Cow::Cow to <property>.
- Adds \$<LINK_ONLY:Cow::Cow> to INTERFACE LINK LIBRARIES.

Pure usage reqiurements

```
add_library(Bar INTERFACE)
target_compile_definitions(Bar INTERFACE BAR=1)
```

- INTERFACE libraries have no build specification.
- They only have usage requirements.

Don't abuse requirements!

Eg: -Wall is not a requirement!

Project Boundaries

How to use external libraries

Always like this:

```
find_package(Foo 2.0 REQUIRED)
target_link_libraries(... Foo::Foo ...)
```

FindFoo.cmake

```
find_path(Foo_INCLUDE_DIR foo.h)
find library(Foo LIBRARY foo)
3 mark_as_advanced(Foo_INCLUDE_DIR Foo_LIBRARY)
4
5 include(FindPackageHandleStandardArgs)
  find package handle standard args(Foo
    REQUIRED_VARS Foo_LIBRARY Foo_INCLUDE_DIR
  if(Foo FOUND AND NOT TARGET Foo::Foo)
    add library(Foo::Foo UNKNOWN IMPORTED)
11
    set target properties(Foo::Foo PROPERTIES
12
      IMPORTED_LINK_INTERFACE_LANGUAGES "CXX"
13
      IMPORTED LOCATION "${Foo LIBRARY}"
14
      INTERFACE_INCLUDE_DIRECTORIES "${Foo_INCLUDE_DIR}"
15
16
17 endif()
```

FindPNG.cmake

```
if(PMG_FIND_QUIETLY)
set(_FIND_ZLIB_ARG QUIET)
endif()
   endif()
find_puckage(ZLIB $(_FIND_ZLIB_ARG))
              find_path(PMS_PMS_DECLUDE_DIR pmg.h
/usr/local/include/tibpng # OpenBSD
              list(APPEND PNG NAMES one libone)
              THIS CONTROL FOR STATE OF THE S
       etts ()

STREET (MELTER SUFFICHIN)

STREET (MELTER SUFFICHIN)

FROM VESTOR SUFFICHIN)

FROM VESTOR SUFFICHIN)

STREET (MELTER SUF
              endif ()
foreach(v IN LISTS _PMG_VERSION_SUFFIXES)
List(APPEND PMG_MAMES png$(v) libpng$(v))
                 emffcreatrs, j
emset_PMc_WRSION_SEFFIES)
# For compatibility with versions prior to this multi-config search, honor
# any PMC_INDRAWY that is already specified and skip the search.
              # any PMS_LIBBARY that is already specified and skip the search. 
if(MOT PMS_LIBBARY) | find_library_RELEASE NAMES ${PMS_MAMES} | find_library_RMS_LIBRARY_RELEASE NAMES ${PMS_MAMES_DEBMS}| include{${CMMS_CURRENT_LIST_DEBY_Selectibrary_Configurations.cnake}}|
          select_library_configurations(PMS)
mark_ms_advanced(PMS_LIBRARY_EXTRAST_PMS_LIBRARY_EXTRAST_PMS_
mark_ms_advanced(PMS_LIBRARY_EXTRAST_PMS_LIBRARY_EXTRAST_PMS_
mass(FMS_MARKS)
              unset(PNG FOUND)
              if (PMG_LIBRARY AND PMG_PMG_INCLUDE_DIR)
                                   # prog.h includes glib.h. Sigh.
set(PMC_INCLIDE_DIRS $[PMC_PMC_INCLIDE_DIR) $ [ZLIB_ENCLUDE_DIR) }
set(PMC_INCLIDE_DIRS $[PMC_PMC_INCLIDE_DIRS) ) # for backward compatibility
set(PMC_LIBERRIES $[PMC_LIBERRY] $ [ZLIB_LIBERRY])
                                             if(BUILD_SHARED_LIBS)
                                   IT(NULL_MONESC_LESS)

# No need to define PMG_USE_DLL here, because it's default for Cygnic.
else()
set (PMG_DEFINITIONS -DPMG_STATEC)
endif()
endif()
                                   if(NOT TARGET PNG::PNG)
add_library(PNG::PNG UNKNOWN IMPORTED)
                                             abs_liprary(west; who descent institute)
set_target_properties(PMG::PMG PROPERTIES
interPACE_compling_printfions) "$[PMG_pacInitions)"
interPACE_compling_printfions "$[PMG_pacInitions]"
interPACE_limi_limpAcinit_rimitions[]
if(Exists "$[PMG_limpAcinit_rimitions])"
if(Exists "$[PMG_limpAcinit_rimitions])"
                                                        if(KISTS '$(PMG_LIBRARY)')
set_target_properties(PMG::PMG PROPERTIES
IMPORTED_LIBK_INTERFACE_LANGUAGES 'C'
IMPORTED_LOCATION '$(PMG_LIBRARY)')
                                             endif()
if(EXISTS "%[PRC_LIBDARY_DELEASE("))
set_property(TARGET PRG_1996_ADPEND PROPERTY
set_target_properties("PRG_1996_ADPEND PROPERTY
set_target_properties("PRG_1996_ADPEND PROPERTIES
1990FTD_LIBG_INTERFACE_LANGUAGES_BELEASE(")
1990FTD_LIBG_INTERFACE_LANGUAGES_BELEASE(")
1990FTD_LIBG_INTERFACE_LANGUAGES_BELEASE(")
1990FTD_LIBG_INTERFACE_LANGUAGES_BELEASE(")
                                                 IMPORTID_COLVETON_NELECTOR 'SPEE_LEBOOK', is
endif()
if(EXISTS '$(PMG_LIBDARY_DEBUG)')
set_property COMPGED_NELECTOR
IMPORTID_COMPGED_NELECTOR

                                                            set_target_properties(PMG::PMG PROPERTIES
IMPORTED_LIMO_INTERFACE_LANGUAGES_DEBUG "C"
IMPORTED_LOCATION_DEBUG "$(PMG_LIBRARY_DEBUG)")
              if (PMG PMG INCLUDE DIR AND EXISTS "$(PMG PMG INCLUDE DIR)/one.h")
                                   file(STRIMGS "$[PMG_PMG_INCLUDE_DIR)/pmg_h" pmg_version_str REGEX ""#define[ \t]-PMG_LIBPMG_VER_STRIMG[ \t]-\".-\"")
                                      string(REGEX REPLACE '*midefine[ \t]-PME_LIEPMS_VER_STRING[ \t]+\'(["\"]+)\'.*' \\1" PMS_VERSION_STRING "$[png_version_str]")
endif()
endif()
   include($(CMAXE_CURRENT_LIST_DIR)/FindPackageHandleStandardArgs.cmake)
   find_package_handle_standard_args(PNG
REQUIRED_WAS PNG_LIERARY PNG_PNG_INCLUDE_DIR
WESSION VAR PNG_WESSION STRING)
   mark as advanced(PNG PNG INCLUDE DIR PNG LIBRARY )
```

Use a Find module for third party libraries that are not built with CMake.

Use a Find module for third party libraries that are not built with CMake.

that do not support clients to use CMake.

Use a Find module for third party libraries that are not built with CMake.
that do not support clients to use CMake.

Also, report this as a bug to their authors.

Export your library interface!

```
find package(Bar 2.0 REQUIRED)
add library(Foo ...)
3 target link libraries(Foo PRIVATE Bar::Bar)
4
5 install(TARGETS Foo EXPORT FooTargets
    ITBRARY DESTINATION lib
    ARCHIVE DESTINATION lib
    RUNTIME DESTINATION bin
    INCLUDES DESTINATION include
10
  install(EXPORT FooTargets
    FILE FooTargets.cmake
12
    NAMESPACE Foo::
    DESTINATION lib/cmake/Foo
15
```

Export your library interface!

```
include(CMakePackageConfigHelpers)
write_basic_package_version_file("FooConfigVersion.cmake"

VERSION ${Foo_VERSION}
COMPATIBILITY SameMajorVersion

install(FILES "FooConfig.cmake" "FooConfigVersion.cmake"

DESTINATION lib/cmake/Foo

)
```

```
include(CMakeFindDependencyMacro)
find_dependency(Bar 2.0)
include("${CMAKE_CURRENT_LIST_DIR}/FooTargets.cmake")
```

Export the right information!

Warning:

The library interface may change during installation. Use the BUILD_INTERFACE and INSTALL_INTERFACE generator expressions as filters.

Creating Packages

CPack

- CPack is a packaging tool distributed with CMake.
- set() variables in CPackConfig.cmake, or
- set() variables in CMakeLists.txt and include(CPack).

Write your own CPackConfig.cmake

that is generated by CMake.

and include() the one

CPack secret

The variable CPACK_INSTALL_CMAKE_PROJECTS is a list of quadruples:

- 1. Build directory
- 2. Project Name
- 3. Project Component
- 4. Directory

Packaging multiple configurations

1. Make sure different configurations don't collide:

```
set(CMAKE_DEBUG_POSTFIX "-d")
```

- 2. Create separate build directories for debug, release.
- 3. Use this CPackConfig.cmake:

```
include("release/CPackConfig.cmake")
set(CPACK_INSTALL_CMAKE_PROJECTS
    "debug;Foo;ALL;/"
    "release;Foo;ALL;/"
    )
```

Package Management

My requirements for a package manager

- Support system packages
- Support prebuilt libraries
- Support building dependencies as subprojects
- Do not require any changes to my projects!

How to use external libraries

Always like this:

```
find_package(Foo 2.0 REQUIRED)
# ...
target_link_libraries(... Foo::Foo ...)
```

• System packages ...

- · System packages ...
 - · work out of the box.

- · System packages ...
 - · work out of the box.
- · Prebuilt libraries ...

- · System packages ...
 - · work out of the box.
- · Prebuilt libraries ...
 - need to be put into CMAKE_PREFIX_PATH.

- · System packages ...
 - · work out of the box.
- · Prebuilt libraries ...
 - · need to be put into CMAKE_PREFIX_PATH.
- · Subprojects ...
 - · We need to turn find_package(Foo) into a no-op.
 - What about the imported target Foo::Foo?

Use the your public interface

```
When you export Foo in namespace Foo:;, also create an alias Foo::Foo.
```

```
add_library(Foo::Foo ALIAS Foo)
```

When you export Foo in namespace Foo::, also create an alias Foo::Foo.

The toplevel super-project

```
set(CMAKE PREFIX PATH "/prefix")
set(as subproject Foo)
3
4 macro(find package)
    if(NOT "${ARGO}" IN LIST as subproject)
      find package(${ARGV})
    endif()
8 endmacro()
9
10 add subdirectory(Foo)
11 add subdirectory(App)
```

How does that work?

If Foo is a ...

- · system package:
 - find_package(Foo) either finds FooConfig.cmake in the system or uses FindFoo.cmake to find the library in the system. In either case, the target Foo::Foo is imported.
- · prebuilt library:
 - find_package(Foo) either finds FooConfig.cmake in the CMAKE_PREFIX_PATH or uses FindFoo.cmake to find the library in the CMAKE_PREFIX_PATH. In either case, the target Foo::Foo is imported.
- · subproject:
 - find_package(Foo) does nothing.
 The target Foo::Foo is part of the project.

CTest

Run with ctest -S build.cmake

```
set(CTEST_SOURCE_DIRECTORY "/source")
2 set(CTEST BINARY DIRECTORY "/binary")
3
4 set(ENV{CXXFLAGS} "--coverage")
5 set(CTEST CMAKE GENERATOR "Ninja")
6 set(CTEST USE LAUNCHERS 1)
7
8 set(CTEST COVERAGE COMMAND "gcov")
9 set(CTEST MEMORYCHECK COMMAND "valgrind")
10 #set(CTEST_MEMORYCHECK_TYPE "ThreadSanitizer")
11
12 ctest_start("Continuous")
13 ctest configure()
14 ctest build()
15 ctest test()
16 ctest coverage()
17 ctest memcheck()
18 ctest submit()
```

CTest scripts are the right place

Keep that information out of the project.

for CI specific settings.

Filtering tests by name

Define like this:

```
add_test(NAME Foo.Test
COMMAND foo_test --number 0
)
```

Run like this:

```
1 ctest -R 'Foo.' -j4 --output-on-failure
```

Follow a naming convention for test names.

This simplifies filtering by regex.

Fail to compile

```
add library(foo fail STATIC EXCLUDE FROM ALL
    foo fail.cpp
4 add test(NAME Foo.Fail
   COMMAND ${CMAKE COMMAND}
      --build ${CMAKE BINARY DIR}
     --target foo fail
 set property(TEST Foo.Fail PROPERTY
    PASS REGULAR EXPRESSION "static assert message"
10
    )
11
```

Running crosscompiled tests

- When the testing command is a build target, the command line is prefixed with \${CMAKE_CROSSCOMPILING_EMULATOR}.
- When crosscompiling from Linux to Windows, set CMAKE_CROSSCOMPILING_EMULATOR to wine.
- When crosscompiling to ARM, set CMAKE_CROSSCOMPILING_EMULATOR to qemu-arm.
- To run tests on another machine, set CMAKE_CROSSCOMPILING_EMULATOR to a script that copies it over and executes it there.

Run tests on real hardware

```
1 #!/bin/bash
2 tester=$1
3 shift
4 # create temporary file
5 filename=$(ssh root@172.22.22.22 mktemp)
6 # copy the tester to temporary file
າ scp $tester root@172.22.22.$filename
8 # make test executable
9 ssh root@172.22.22.22 chmod +x $filename
10 # execute test
ssh root@172.22.22.22 $filename "$@"
12 # store success
13 success=$?
14 # cleanup
15 ssh root@172.22.22.22 rm $filename
16 exit $success
```

Cross Compiling

Toolchain.cmake

```
set(CMAKE SYSTEM NAME Windows)
2
₃ set(CMAKE C COMPILER
                         x86 64-w64-mingw32-gcc)
4 set(CMAKE CXX COMPILER x86 64-w64-mingw32-g++)
5 set(CMAKE RC COMPILER
                         x86 64-w64-mingw32-windres)
6
7 set(CMAKE FIND ROOT PATH /usr/x86_64-w64-mingw32)
8
9 set(CMAKE FIND ROOT PATH MODE PROGRAM NEVER)
 set(CMAKE_FIND_ROOT_PATH_MODE_LIBRARY ONLY)
  set(CMAKE_FIND_ROOT_PATH_MODE_INCLUDE ONLY)
12
13 set(CMAKE CROSSCOMPILING EMULATOR wine64)
```

Don't put logic in toolchain files.

Static Analysis

Treat warnings as errors?

How do you treat build errors?

- · You fix them.
- · You reject pull requests.
- · You hold off releases.

Treat warnings as errors!

Treat warnings as errors!

- · You fix them.
- · You reject pull requests.
- · You hold off releases.

Treat warnings as errors!

To treat warnings as errors, never pass -Werror to the compiler. If you do, your compiler treats warnings as errors. You can no longer treat warnings as errors, because you will no longer get any warnings. All you get is errors.

-Werror causes pain

- You cannot enable -Werror unless you already reached zero warnings.
- You cannot increase the warning level unless you already fixed all warnings introduced by that level.
- You cannot upgrade your compiler unless you already fixed all new warnings that the compiler reports at your warning level.
- You cannot update your dependencies unless you already ported your code away from any symbols that are now [[deprecated]].
- You cannot [[deprecated]] your internal code as long as it is still used. But once it is no longer used, you can as well just remove it...

Better: Treat new warnings as errors!

- 1. At the beginning of a development cycle (eg. sprint), allow new warnings to be introduced.
 - · Increase warning level, enable new warnings explicitly.
 - · Update the compiler.
 - · Update dependencies.
 - Mark symbols as [[deprecated]].
- 2. Then, burn down the number of warnings.
- 3. Repeat.

Pull out all the stops!

- **clang-tidy** is a clang-based C++ "linter" tool. Its purpose is to provide an extensible framework for diagnosing and fixing typical programming errors, like style violations, interface misuse, or bugs that can be deduced via static analysis.
 - **cpplint** is automated checker to make sure a C++ file follows Google's C++ style guide.
- include-what-you-use analyzes #includes in C and C++ source
 files.
 - **clazy** is a clang wrapper that finds common C++/Qt antipatterns that decrease performance.

Target properties for static analysis

- · <lang>_CLANG_TIDY
- · <lang>_CPPLINT
- · <lang>_INCLUDE_WHAT_YOU_USE
 - · Runs the respective tool along the with compiler.
 - · Diagnostics are visible in your IDE.
 - · Diagnostics are visible on CDash.
- · LINK WHAT YOU USE
 - links with -Wl, --no-as-needed, then runs ldd -r -u.

<lamp> is either C or CXX.

Each of those properties is initialzied with CMAKE_<property>.

Scanning header files

- Most of those tools report diagnostics for the current source file plus the associated header.
- · Header files with no assiciated source file will not be analyzed.
- You may be able to set a custom header filter, but then the headers may be analyzed multiple times.

For each header file, there is an associated source file that #includes this header file at the top.

Even if that source file would otherwise be empty.

Create associated source files

Enable warnings from from outside the project

```
env CC=clang CXX=clazy cmake \
   -DCMAKE_CXX_CLANG_TIDY:STRING=\
   'clang-tidy;-checks=-*,readability-*' \
   -DCMAKE_CXX_INCLUDE_WHAT_YOU_USE:STRING=\
   'include-what-you-use;-Xiwyu;--mapping_file=/iwyu.imp' \
   ...
```

Supported by all IDEs

- Just setting CMAKE_CXX_CLANG_TIDY will make all clang-tidy diagnostics appear in your normal build output.
- · No special IDE support needed.
- If IDE understands fix-it hints from clang, it will also understand the ones from clang-tidy.



Personal Wishlist

Personal Wishlist

- For each of the following ideas, I have started a prototype.
- · Contibutions welcome!
- · You can talk to me!

Disclaimer:

No guarantee that the following ideas will ever be added to CMake.

PCH as usage requirements

PCH as usage requirements

```
target_precompile_headers(Foo
PUBLIC
"foo.h"
PRIVATE
vunordered_map>
)
```

PCH as usage requirements

- Calculate a list of headers per config and language for the build specification of each target.
- · Generate a header file that **#include**s each of those headers.
- Tell the build system to precompile this header.
- Tell the build system to force-include this header.
- · Require no changes to the code (No #include "stdafx.h").

More Languages!

More Languages!

- · CMake's core is language-agnostic.
- · Language support is scripted in modules.
- · Rules how to compile object files, link them together.
- The output of the compiler must be an object file.
- CMake can be used with D by putting necessary files in CMAKE_MODULE_PATH.¹

¹https://github.com/dcarp/cmake-d

Even more Languages!

- If we allow the output to be a source file of a known language, we would not need special handling for Protobuf, Qt-resources, or any other IDL.
- This would also allow using CMake for BASIC, BCX, Chapel, COBOL, Cython, Eiffel, Genie, Haxe, Java, Julia, Lisaac, Scheme, PHP, Python, X10, Nim, Vala.²

²https://en.wikipedia.org/wiki/Source-to-source_compiler

find_package(Foo PKGCONF)

find_package(Foo PKGCONF)

- find_package() has two modes: PACKAGE and CONFIG.
- · Let's add a PKGCONF mode.
- In this mode, CMake parses .pc files and generates one IMPORTED library per package.

Declarative Frontend and Lua VM

Lua VM

- Execute CMake commands on Lua VM.3
- · Allow CMake modules to be written in Lua.

³not the other way round. This failed before.

Declarative Frontend

- For directories, use a declarative language that allows procedural subroutines.
- · libucl⁴ is an interesting option.

⁴https://github.com/vstakhov/libucl

Tell me your ideas!